

第 2 章

表計算ソフトウェア：少し複雑な処理 (lookup 関数, 絶対参照)

2022 年 8 月 31 日

学習目標

- (1) 絶対参照の考え方を理解する.
- (2) IF 関数を使って, 条件分けをできるようになる.
- (3) 表引き (LOOKUP 関数) の考え方を学習する.
- (4) さまざまな関数 (ROUNDUP, ABS など) の使い方に慣れる.
- (5) 少し複雑な計算式になれる
- (6) 絶対参照, 相対参照が混じったものに慣れる
- (7) 大きな表を作成してみる

本章は, 専修大学商学部の高萩栄一郎の著作である.

1 表計算ソフトウェアのしくみ

計算式を複製するとき、複製元の計算式中セルの番地は、複製に伴う番地の移動に伴い、複製先の計算式中の番地は自動的に変化します。絶対参照（または絶対番地）とは、複製に伴いセルが移動しても変化させない（固定する）番地の指定方法です。逆に、セルの移動とともに変化させるセルの番地の指定を相対参照（または相対番地）といいます。プログラミングとの対応では、ループの中での定数が絶対参照、ループの中の配列の要素が相対参照に対応します。

2 複製のしくみ

図 1 は、GDP の表を複製で設定した計算式で、次のことがわかります。

S2:T12(図 1 の青の部分) 式がかかっているセルの数字が 1 増えると、計算式中の数字も 1 増えています。たとえば、S2 の計算式は「=SUM(B2:R2)」になっており、下に複製した S3 の計算式は「=SUM(B3:R3)」と、すべて数字が 1 増えています。

B12:R13(図 1 の赤の部分 (一部)) 式が書かれているセルのアルファベットが 1 つ増えると、右に複製された計算式のアルファベットも 1 つ増えています。たとえば、P13 の計算式は「=AVERAGE(P2:P11)」になっており、右に複製した Q13 の計算式は、「=AVERAGE(Q2:Q11)」とアルファベットが 1 つ増えています。

したがって、「複製」という機能は、複製元のセルの番地と複製先のセルの番地との差の分、計算式中の番地を調整していることがわかります。別の言い方をすれば、計算式は設定しているセルからの相対的な位置で記録され、これは相対参照と呼ばれています。たとえば、図 1 の S2 の計算式は、「=SUM(B2:R2)」になっています。この計算式は、

=SUM((自分より 17 つ左のセル)から(自分より 1 つ左のセル)まで)

	A	P	Q	R	S	T
1	Country	2015	2016	2017	合計	平均
2	Australia	1232.913	1264.524	1379.548	=SUM(B2:R2)	=AVERAGE(B2:R2)
3	Canada	1559.623	1535.768	1653.043	=SUM(B3:R3)	=AVERAGE(B3:R3)
4	China	11226.186	11221.836	12014.61	=SUM(B4:R4)	=AVERAGE(B4:R4)
5	India	2102.392	2273.556	2602.309	=SUM(B5:R5)	=AVERAGE(B5:R5)
6	Japan	4394.977	4950.073	4873.202	=SUM(B6:R6)	=AVERAGE(B6:R6)
7	Korea	1382.764	1414.804	1540.458	=SUM(B7:R7)	=AVERAGE(B7:R7)
8	United Kingdom	2897.06	2669.107	2628.41	=SUM(B8:R8)	=AVERAGE(B8:R8)
9	United States	18219.3	18707.15	19485.4	=SUM(B9:R9)	=AVERAGE(B9:R9)
10	EURO area	11686.259	11960.698	12632.374	=SUM(B10:R10)	=AVERAGE(B10:R10)
11	Asean-5	2047.966	2148.331	2317.146	=SUM(B11:R11)	=AVERAGE(B11:R11)
12	合計	=P2+P3+P4+P5+P6+P	=Q2+Q3+Q4+Q5+Q6+Q	=R2+R3+R4+R5+R6+R	=SUM(B12:R12)	=AVERAGE(B12:R12)
13	平均	=AVERAGE(P2:P11)	=AVERAGE(Q2:Q11)	=AVERAGE(R2:R11)		

複写すると数字が増えていく

複写するとアルファベットが増えていく

図1 相対参照のしくみ (計算式を表示)

というしくみになっています。多くの表は、この複写の番地調整機能（相対参照）でうまくいくのですが、うまくいかないことがあります。その例をやってみましょう。

■参考:計算式の表示方法 計算式を設定するとその計算結果（数値など）が表示されます。計算結果に替えて、計算式を表示することもできます。

- メニューの 数式 → ワークシート分析の中の 数式の表示
- もう一度 数式の表示 をクリックすることにより元に戻る

3 絶対参照と相対参照

図 2 は、各支店の年度別の売上高を記入したものです。この表は、教科書のページからダウンロードできます (S6.xlsx)。オレンジの背景色の部分は、計算式で入力する部分です

	A	B	C	D	E	F	G
1		A支店	B支店	C支店	D支店	E支店	合計
2	売上高(2010)	1200	1800	2200	1100	1200	7500
3	売上高(2011)	1300	1800	2100	1200	1250	7650
4	伸び率	8%	0%	-5%	9%	4%	2%
5	構成比率(2011)	17%	24%	27%	16%	16%	100%
6	伸び率評価1	1	0	0	1	1	
7	伸び率評価2	1	0	0	1	1	

図 2 売上高の評価

伸び率を計算しましょう。伸び率の計算式は、

$$\frac{2011 \text{ 年の売上高} - 2010 \text{ 年の売上高}}{2010 \text{ 年の売上高}}$$

で、表示形式は「パーセント」です。

3.1 絶対参照

次に、2011 年の各支店の売り上げが、合計に占める割合（構成比率）を計算しましょう。

$$\times \boxed{B5:} = B3/G3$$

となります。この計算式を B5 から F5 に、複製して見ましょう。おかしい結果になります。原因は表 1 の「単純に複製した

ときの計算式」(相対参照)になってしまうからです。

セル	B5	C5	D5	E5	F5
単純に複製したときの計算式	=B3/G3	=C3/H3	=D3/I3	=E3/J3	=F3/K3
正しい計算式	=B3/G3	=C3/G3	=D3/G3	=E3/G3	=F3/G3




図3 絶対参照を使って設定した計算式

C5 から F5 の計算式がおかしいことがわかります。分子は複製先のセルの番地に従って、セルの番地は、相対参照(番地調整機能)によってアルファベットが1つずつ増えてよいのですが、分母の番地はいつも「G3」のままでなくてはなりません(青)。そこで、複製元(B5)のセルを設定するとき、番地調整機能を働かせないようにします。これは、変化させないアルファベットや数字の前に、「\$」マークをつけます。したがって、複製元の計算式は、次式になります。

B5: =B3/\$G3

この場合、G3 は、常に G3 で、数値の方は変化しないので、

B5: =B3/\$G\$3

としてもかまいません。\$をつけた番地を「絶対参照」、\$を付けない番地を「相対参照」といいます (図 3)。

■絶対参照の指定の仕方 他の計算式と同様に、キーボードなどから\$マークを入力することが可能ですが、セルをクリック (タップ) したあとに、簡単な操作で\$マークを入力する方法があります。

- セルの番地 (G3 など) をクリックなどで入力します。
- キーボードの(F4)キーを押すことにより、\$マークを付けることができます。
- \$ の位置が異なるときは、(F4)キーを何回か押すことにより、指定の計算式にします。

[動画:絶対参照 \(音声付き\)](#) ・ [動画:絶対参照 \(音声なし\)](#)

3.2 スピルを使った指定 (参考)

Excel 2021 で追加された スピル という機能で、絶対参照を使って指定した範囲の計算式をまとめて指定することができます。この機能は、2022 年度端末室に導入されている Excel 2019 では利用できません。ただし、オンライン版の Excel では利用できます。

■スピルによる指定

- (1) まとめて指定する範囲の左上のセルを選択します (B5)
- (2) 次の計算式を入力して、**Enter**

B5: =B3:F3/G3

- (3) B3:F5 は、B3 から右へ 5 個のセルなので、計算式を入力した B5 から F5 までの 5 個のセルに こぼれる (spill) ように、計算結果が表示されます。

スピルは、計算式 (B5) により、出力範囲を拡張 (B5:F5) して、計算結果を表示します。

■注意

- 「#SPILL」というエラーが表示されることがあります。これは、拡張された出力範囲にすでに値が入力されている場合に表示されます。
- スピルで設定した計算式を変更するときは、左上のセルの計算式を変更します。

[動画:スピル \(音声付き\)](#) ・ [動画:スピル \(音声なし\)](#)

4 指数を計算

GDP の計算で、経済発展が大きい国はどこであるのか分析するとき、国の大きさに関わらず、GDP がどのように変化しているのかを分析することが重要です。そこで、物価指数のように基準年を設け、各国について基準年の値を 100 としたとき各年の GDP がいくつかになるのかを分析します。

	A	B	C	D	E	F	G	H	I
1	GDP指数 2001年基準	2001	2002	2003	2004	2005	2006	2007	2008
2	Australia	100.00	112.65	143.24	174.14	194.84	207.22	251.38	279.93
3	Canada	100.00	102.93	121.19	138.94	158.80	178.63	198.94	210.36
4	China	100.00	109.93	124.33	146.29	171.78	206.41	265.72	342.56
5	India	100.00	106.04	125.19	146.09	168.89	192.15	250.78	247.83
6	Japan	100.00	95.62	103.30	111.89	110.50	105.27	104.92	117.06
7	Korea	100.00	114.25	127.66	143.49	168.49	189.81	210.61	188.02
8	United Kingdom	100.00	109.20	125.79	147.87	155.44	166.09	189.72	180.46
9	United States	100.00	103.35	108.28	115.42	123.20	130.55	136.57	139.04
10	EURO area	100.00	109.08	134.40	154.05	159.99	169.76	195.69	215.39
11	Asean-5	100.00	113.64	129.07	144.53	160.78	193.70	230.50	270.35
12	全体	100.00	104.77	117.60	130.84	138.86	147.80	164.10	175.99

図4 指数化した表

図4は、その計算表で新しく表（シート名：GDP_index）を作成しました。計算式は、2001年のデータを100として、その何倍になったかを計算するので、

$$\text{A国 t 年の指数 (2001年基準)} = \frac{\text{A国 t 年の GDP}}{\text{A国 2001 年の GDP}} \times 100$$

となります。

B2の計算式は、ワークシート間の演算を利用して「=(GDP!B2/GDP!B2)*100」ですが、複写するとき、2000年のGDPはいつもB列なので、Bを変化させないようにBの前に\$をつけ2の前には付けません。(F4)キーを使うとき分母のB2をクリックしたあと、(F4)キーを3回押すことにより!B2になります。2000年から2007年までまとめて、計算式を複写します。

B2:	=(GDP!B2/GDP!\$B2)*100
複写元:	B2
複写先:	B2:O13

とします。B列も同じ計算式で、100となります。ⓕ4キーは、1回押す毎に\$マークが付く位置（アルファベットの前、数字の前）を変更しますので、何回か押して正しい位置になるようにしてください。 [動画:Excel GDP の指数](#)

■スピルによる計算（参考）

- ワークシート GDP_index を選択
- まとめて指定する範囲の左上のセルを選択します (B2)
- B2 にⓓを入力
- GDP のワークシートで、B2:R11 を範囲指定し、Ⓛを入力
- GDP のワークシートで、B2:B11 を範囲指定
- 「*100」を入力し、ⓔnter
- 計算式は次のようになります。

B2:	=GDP!B2:R11/GDP!B2:B11*100
-----	----------------------------

- 計算式で「GDP!B2:R11」とあるので、B列からR列まで17列、2行目から11行目まで10行、式を入力するB2から右下へ17列11行まとめて計算します。
- 分母の「GDP!B2:B11」は行のみの変化ですので、行のみ変化させて計算します。
- B2:B11が、2001年基準（B列）を表します。

[動画:GDP 指数スピル（音声付き）](#) ・ [動画:GDP 指数（音声なし）](#)

4.1 練習問題 S1-6

- (1) シート「GDP_index」を作成し、2008年基準（リーマンショック）のGDPの指数の表を作成しなさい。
- (2) シート「population_index」を作成し、2001年基準の人口の指数の表を作成しなさい。
- (3) シート「GDPperCapita_index」を作成し、2010年基準の一人あたりGDPの指数の表を作成しなさい。

5 練習問題

5.1 練習問題 S7-1

図5は、関東の各都県の小学生、中学生、高校生、大学生の人数を千人単位で記入したものです。（出典：学校基本調査、1999年5月1日現在）この表は、[教科書のページ](#)からダウンロードできます（S7.xlsx）。

	A	B	C	D	E	F	G	H	I
1		人数(単位:1000人)				各都県が占める割合			
2	都県名	小学生	中学生	高校生	大学生	小学生	中学生	高校生	大学生
3	茨城	189	110	106	37	8.55%	8.94%	8.58%	3.20%
4	栃木	128	74	77	21	5.79%	6.01%	6.23%	1.82%
5	群馬	124	69	67	19	5.61%	5.61%	5.42%	1.65%
6	埼玉	408	217	206	114	18.45%	17.63%	16.67%	9.87%
7	千葉	339	190	183	114	15.33%	15.43%	14.81%	9.87%
8	東京	562	329	365	660	25.42%	26.73%	29.53%	57.14%
9	神奈川	461	242	232	190	20.85%	19.66%	18.77%	16.45%
10	合計	2211	1231	1236	1155	100.00%	100.00%	100.00%	100.00%

図5 関東の県の児童生徒学生の比率

- (1) B10 から E10 に、小学生、中学生、高校生、大学生の人数の合計を計算しなさい。
- (2) F3 に、茨城県の小学生人数の関東の都県の合計に占める割合（パーセント表示）を計算します。ただし、F3 の計算式を F3 から F9 に複写するために、F3 の計算式の適当な位置に\$マークをつけます。ただし、\$マークは変化させない

ものの前につけます。アルファベットを変化させたくない場合は、アルファベットの前、数字を変化させたくない場合は数字の前に\$をつけます。

- (3) F3 の計算式を F3 から F9 に複製します。
- (4) F10 に各県の割合の合計を計算します。この値は、100%（または 1）になるはずです。
- (5) 中学生、高校生、大学生の列も同様に計算式を設定します。

5.2 練習問題 S8-1

図 6 のように入試の合否判定をします。この表は、[教科書のページ](#)からダウンロードできます (S8.xlsx)。ただし、E 列の合否判定は、IF 関数の節で学習します。

- (1) 得点の合計、平均を計算しましょう。
- (2) 平均との差を計算しましょう。D6 に、平均との差を求める式を下に複製することを考えて作成します。
平均値はいつも C17 であることを注意しましょう。

	A	B	C	D	E
1	合否判定				
2	合格基準点	70			
3					
4					
5	番号	名前	得点	平均との差	合否
6	1	田中	100	34.8	1
7	2	大西	70	4.8	1
8	3	大久保	30	-35.2	0
9	4	前田	80	14.8	1
10	5	友田	65	-0.2	0
11	6	保田	77	11.8	1
12	7	大川	78	12.8	1
13	8	木村	32	-33.2	0
14	9	村田	54	-11.2	0
15	10	川上	66	0.8	0
16		合計	652		5
17		平均	65.2		

図6 合否判定

5.3 練習問題 S1-7

各国・地域の GDP が各年，10 ヶ国・地域の合計に対しての割合（構成割合）を求めましょう。

(1) 新しいシート「GDP_share」を作成し，指数と同様に，表頭，表側を複写します。

(2) A 国・地域の t 年の構成割合は、次式で計算します。

$$\text{A 国・地域 t 年の構成割合} = \frac{\text{A 国・地域 t 年の GDP}}{\text{t 年の 10 ヶ国・地域の GDP の合計}}$$

B2 に計算式を設定します。B2 の場合、分子は 2001 年の Australia の GDP(GDP!B2) になります。分母は、2001 年の GDP の合計 (GDP!B12) になりますが、複写するとき、合計はいつも 12 行目であることに注意しましょう。

(3) 計算式を複写します。

複写元: B2 複写先: B2:R11

(4) 表示形式をパーセンテージにします。

(5) 図 7 は、計算例です。

	A	B	C	D	E	F	G	H
1	GDP構成割合	2001	2002	2003	2004	2005	2006	2007
2	Australia	1.39%	1.49%	1.69%	1.85%	1.95%	1.95%	2.13%
3	Canada	2.72%	2.67%	2.80%	2.89%	3.11%	3.29%	3.30%
4	China	4.96%	5.21%	5.25%	5.55%	6.14%	6.93%	8.04%
5	India	1.82%	1.85%	1.94%	2.04%	2.22%	2.37%	2.79%
6	Japan	15.89%	14.50%	13.96%	13.59%	12.64%	11.32%	10.16%
7	Korea	1.97%	2.15%	2.14%	2.16%	2.39%	2.53%	2.53%
8	United Kingdom	6.00%	6.26%	6.42%	6.79%	6.72%	6.75%	6.94%
9	United States	39.07%	38.54%	35.97%	34.46%	34.66%	34.51%	32.51%
10	EURO area	24.32%	25.32%	27.79%	28.63%	28.02%	27.93%	29.00%
11	Asean-5	1.86%	2.02%	2.04%	2.05%	2.15%	2.43%	2.61%

図 7 GDP の構成割合 (一部)

6 IF 関数

6.1 IF 関数とは

S6 の売上高評価の表で、「伸び率評価1」の計算式を設定しましょう。これは、伸び率が正だったら 1, 0 または負だったら, 0 を表示させます。

	A	B	C	D	E	F	G
1		A支店	B支店	C支店	D支店	E支店	合計
2	売上高(2010)	1200	1800	2200	1100	1200	7500
3	売上高(2011)	1300	1800	2100	1200	1250	7650
4	伸び率	8%	0%	-5%	9%	4%	2%
5	構成比率(2011)	17%	24%	27%	16%	16%	100%
6	伸び率評価1	1	0	0	1	1	
7	伸び率評価2	1	0	0	1	1	

図 8 売上高の評価(再掲)

あるときは、こうして、そうでないときは、こうするといったことをしたいときは、IF 関数を使います。使い方は、

IF(条件, 条件を満たしているときすること, 条件を満たしていないときすること)

となります。したがって、

B6: =IF(B4>0,1,0)

とします。その意味は、図 9 のようになります。




図9 IF関数の図解

表2 Excelでの大小関係の表記

大小関係	Excelの表記
=	=
≠	<>
>	>
≥	>=
<	<
≤	<=

大小関係などの条件の表記は、数学の表記とは若干異なり、表2のようになります。

練習として、伸び率評価2の計算式を設定しましょう。伸び率評価2は、その支店の伸び率が全体の伸び率（セルG4の値）より大きいまたは等しければ1を、そうでなければ0を表示させます。絶対参照を使いますので注意してください。

6.2 講演会参加費

図 10 は、ある講演会の会費の計算をしたものです。この表は、[教科書のページ](#)からダウンロードできます (S9.xlsx)。

	A	B	C	D	E	F	G	H
1	NO	会員／非会員	学生／非学生	パーティ	参加費	学生割引	パーティ代	合計
2	1	M	S	P	6,000	3,000	4,000	7,000
3	2		S		9,000	3,000	0	6,000
4	3			P	9,000	0	4,000	13,000
5	4	M		P	6,000	0	4,000	10,000
6	5	M	S		6,000	3,000	0	3,000
7	6	M	S		6,000	3,000	0	3,000
8	7	M	S		6,000	3,000	0	3,000
9	8	M		P	6,000	0	4,000	10,000
10	9				9,000	0	0	9,000
11	10	M			6,000	0	0	6,000
12	11	M		P	6,000	0	4,000	10,000
13	12	M			6,000	0	0	6,000
14	13	M			6,000	0	0	6,000
15	14	M	S	P	6,000	3,000	4,000	7,000
16	15	M	S		6,000	3,000	0	3,000
17	16	M		P	6,000	0	4,000	10,000
18	17			P	9,000	0	4,000	13,000
19	18				9,000	0	0	9,000
20	19	M		P	6,000	0	4,000	10,000
21	20	M		P	6,000	0	4,000	10,000
22				合計	135,000	21,000	40,000	154,000
23								
24		参加費(会員)	6,000					
25		参加費(非会員)	9,000					
26		学生割引	3,000					
27		パーティ	4,000					

図 10 講演会会費

会員／非会員 (B 列) 会員と非会員では、参加費が異なります。会員には、「M」と入力されています。

学生／非学生 (C 列) 学生には、参加費が割引（学生割引）されます。学生には、「S」と入力されています。

パーティ (D 列) 講演会終了後に開かれるパーティに参加する人は、「P」と入力されています。

参加費 (E 列) 参加費を記入します。参加費は変更されることもあるので、C24 もしくは C25 を参照するようにします。

学生割引 (F 列) 学生割引を記入します。学生割引は変更されることもあるので、C26 を参照するようにします。

パーティ代 (G 列) パーティ代を記入します。パーティ代は変更されることもあるので、C27 を参照するようにします。

合計 参加費－学生割引＋パーティ代 で計算します。

■参加費の計算 E2 のセルの計算式を設定します。参加費は、B2 が、「M」か空白によって異なるので IF 関数を使います。

$$\triangle \boxed{E2:} = \text{IF}(B2="M",6000,9000)$$

B2 が「M」かどうか調べるのに、「B2="M"」とします。文字は、""で囲みます。M のとき、6000 なので、真の時の値を 6000 とし、偽のときの値を 9000 としました。しかし、その 6000 と 9000 という値は、変更されることがあり、C24 と C25 を参照するようにします。C24,C25 は、下に複写されたときも固定される番地なので、絶対参照にします。したがって、

$$\boxed{E2:} = \text{IF}(B2="M",\$C\$24,\$C\$25)$$

とします。

■参加費の計算（スピルの利用）

$$\boxed{E2:} = \text{IF}(B2:B21="M",C24,C25)$$

6.3 練習問題 S9-1

S9(参加費の計算)を完成させましょう。

6.4 練習問題 S8-2

S8の合否を設定します。合格表示という列に、基準点以上であれば1を、そうでなければ0を表示するようにします。もちろん、合否の列は1箇所だけ計算式を作成して、あとは複製で計算式を設定してください。合格表示の合計を算出すれば合格者数を求めることができます。

基準点を変えて、何人合格になるのか試してみようと思います。つまり、基準点を変えてみて、合格者が何人になるのかシミュレーションを行います。

7 表引き (LOOKUP 関数)

コンピュータを利用しているとよくコードを使います。例えば、大学生なら学生コード、商品には商品コードが付けられています。また、ある範囲での対応表から表を引くという作業があります。例えば、所得額から、税額を求めるといったことです。これらの機能を実現するのがVLOOKUP(または、HLOOKUP)という関数です。

JR 南武線(路線図)の運賃計算の例で説明しましょう。図 11 は、南武線の2つの駅名を入力して、川崎からの距離を求め、その差から、2 駅間の距離を求め、それから運賃を求めるというものです。この表は、[テキストのページ](#)からダウンロードできます(S101.xlsx(運賃計算)、南武線は、電車特定区間の運賃)。

図 11 の表は3つの表からできています。一番上が運賃を計算する表、二番目が駅名と川崎からの距離の対応表、三番目は、距離と運賃の対応表です。手順は次のようになります。

	A	B	C
1		駅名	川崎からの距離
2	乗車駅	稲田堤	
3	降車駅	登戸	
4	2駅間距離 切り上げ前		
5	2駅間距離 切り上げ後		
6	運賃(ICカードを利用)		
7	運賃(切符を利用)		
8			
9	駅名	川崎からの距離	
10	川崎	0	
11	武蔵小杉	7.5	
12	武蔵溝ノ口	12.7	
13	登戸	17.3	
14	稲田堤	20.8	
15	稲城長沼	24.1	
16	府中本町	27.9	
17	分倍河原	28.8	
18	立川	35.5	
19			
20	距離	運賃(IC)	運賃(切符)
21	1	133	140
22	4	154	160
23	7	165	170
24	11	216	220
25	16	302	310
26	21	388	390
27	26	464	470
28	31	550	550
29	36	637	640
30	41		
31			

運賃を
計算

駅名と川
崎からの
距離の対
応表

距離と
運賃の
対応表

図 11 JR 南武線の運賃計算

- (1) 乗車駅名から川崎からの距離を求めます (C2).
- (2) 降車駅名から川崎からの距離を求めます (C3).
- (3) 2つの「川崎からの距離」の差をもとめ、小数点未満を切り上げます (C4,C5).
- (4) 距離と運賃の対応表から運賃をもとめます (C6,C7).

7.1 乗車距離から運賃を求める

まず乗車距離から運賃 (C6) を求めてみましょう。仮に乗車距離として C5 に 4 を入力します。C6 を求めるには、距離と運賃の対応表を C5 の 4 について引くことです。乗車距離は 4 なので表の口の行「4 154」の行があてはまり運賃は 154 円になります。

JRの運賃表			Excelの表		
	距離	運賃(IC)	距離	運賃(IC)	
イ	1~3	133	1	133	←1以上4未満のとき133円
ロ	4~6	154	4	154	(距離は整数なので1~3のとき130円)
ハ	7~10	165	7	165	
ニ	11~15	216	11	216	←11以上16未満のとき216円
ホ	16~20	302	16	302	(距離は整数なので11~15のとき216円)
ヘ	21~25	388	21	388	
ト	26~30	464	26	464	
チ	31~35	550	31	550	
リ	36~40	637	36	637	←36以上41未満のとき637円
			41		(距離は整数なので36~40のとき637円)

図 12 JR 運賃表と Excel の VLOOKUP 関数用の表との対応関係

7.1.1 対応表の見方

図 12 は、JR 運賃表を Excel の VLOOKUP 用の表には、どのように記述したらよいのかを示しています。Excel の表の各行がどの数値をあてはまるかは、その行の距離の値から、次の行の距離の値までです。正確に言うと対応表の各行にあてはまる数値は、その行の距離から次の行の値未満までです。

距離が 1 の行（イ） 距離が 1 以上 4 未満の運賃が 133 円であることを示しています。

距離が 4 の行（ロ） 距離が 4 以上 7 未満の運賃が 154 円であることを示しています

例えば、乗車距離が 22 であつたら、「21 388」の行（へ）にあてはまります。




図 13 VLOOKUP 関数の書き方

7.2 VLOOKUP 関数

C6 の計算式は、図 13 は、VLOOKUP 関数の図解です。

```
C6: =VLOOKUP(C5,$A$21:$C$30,2,TRUE)
```

VLOOKUP 関数は、VLOOKUP(照合値, 照合範囲, 列位置, 照合方法) の 4 つの引数を書きます。(C5 に何も書いていないとエラーになるので、C5 に仮の 2 駅間距離を入力し、この値をいろいろ変えてみてください。)

照合値 表引きをする表で照合する値。この例では、乗車距離 (C5)
(距離と運賃の対応表の一番左の列から探す値)

照合範囲 (対応表の範囲) 照合する表、ただし、照合値の列は、照合する範囲の一番左の列になるようにします。

列位置 照合した行の左から何番目の列を表示するか。一番左の列を 1 列目と数えます。

照合方法 完全に一致するもののみを探すのか (FALSE)、下の行と未満のものも探すか (TRUE) を指定します。

照合方法が TRUE のとき、表引きをする表は、一番左の列をソートキーに昇順 (小さい順) にソートされてなくてはなりません。理由は、照合するとき、照合値が、一番左の列の値と次の行の一番左の列の値の間にあるかどうかで、照合するかしないか決めているからです。

また、照合範囲に「\$」を付けているのは、複製したとき、照合範囲の位置がずれないようにするためです。

※ C7 に、切符を利用した場合の運賃を計算しましょう。3 列目になることに注意しましょう。

表3 切り上げ, 四捨五入, 切り捨て

ROUNDUP(3.14,0) → 4	ROUNDUP(586,-1) → 590	ROUNDUP(16.58,1) → 16.6
ROUND(3.14,0) → 3	ROUND(586,-1) → 590	ROUND(16.58,1) → 16.6
ROUNDDOWN(3.14,0) → 3	ROUNDDOWN(586,-1) → 580	ROUNDDOWN(16.58,1) → 16.5

7.3 駅名から距離を求める

次に、駅名から「川崎からの距離」を求める関数を求めましょう。駅名から距離を求めるときは、完全に一致する行のみを探します。したがって、ソートしておく必要はありません。C2の計算式は、B2を照合値として、\$A\$10:\$B\$18の照合範囲を探し、1列目と等しければ、2列目の値を返すというものです。完全に一致する行のみ探すので、「FALSE」にします。

C2: =VLOOKUP(B2,\$A\$10:\$B\$18,2,FALSE)

7.4 2 駅間距離を求める（絶対値, 切り上げ）

2つの駅間の距離を求めるには、2つの駅の川崎からの距離の差を求めます。負の値になることがあるので、絶対値をとります。絶対値をとる関数は、ABS です。まず、2 駅間距離切り上げ前は、

C4: =ABS(C2 - C3)

となります。JRの運賃計算は、小数点以下の端数は切り上げです。そこで、C4の値を切り上げて、C5に表示します。

C5: =ROUNDUP(C4,0)

となります。ROUNDUP(A,B)で、Aは切り上げる数、Bは切り上げる単位を示しています。0で小数点の位置、1で0.1単位、2で0.01単位、-1で10の単位で切り上げです。同様に、ROUNDが四捨五入、ROUNDDOWNが切り捨てです(表3)。

8 練習問題

8.1 練習問題 S102-1

図 14 は、小田急ロマンスカーの新宿からの運賃、特急料金の表です (路線図)。B14 に駅名を入力したら、B15 の運賃、B16 の特急料金を計算する式を設定しましょう。この表は、[テキストのページ](#)からダウンロードできます (S102.xlsx(ロマンスカー))。

ヒント：B16 を計算するとき、A3:C11 までを照合範囲として、Vlookup 関数を使います。列位置を設定することで特急料金を表示させます。

	A	B	C
1	新宿からのロマンスカー運賃・特急料金		
2		運賃(切符)	特急料金
3	向ヶ丘遊園	250	300
4	新百合ヶ丘	310	410
5	町田	370	410
6	相模大野	370	410
7	本厚木	500	570
8	秦野	670	620
9	新松田	780	690
10	小田原	880	890
11	箱根湯本	1190	890
12			
13	新宿からロマンスカーを利用(切符)		
14	降車駅	秦野	
15	運賃		
16	特急料金		
17	合計		

図 14 ロマンスカーの運賃・料金

8.2 練習問題 S4-4

S4.xlsx のモーグルで、「エア点」の計算と「スピード」の計算では、小数第 3 桁以下を切り捨て、小数点以下第 2 位まで求めるように変更しましょう。

8.3 練習問題 S103-1

図 16 は、表 15 にしたがって、得点から、S,A+,A,B+,B,C+,C,F の評価を計算する表です。VLOOKUP で使う得点と評価の対応表（照合範囲）を作成し、C2~C11 の計算式を設定しなさい。この表は、[テキストのページ](#)からダウンロードできます (S103.xlsx(成績評価))。

得点	評価
90~100	S
85~89	A+
80~84	A
75~79	B+
70~74	B
65~69	C+
60~64	C
0~59	F

図 15 大学の成績評価

	A	B	C	D
1	名前	得点	評価	
2	田中	100	S	
3	大西	70	B	
4	大久保	30	D	
5	前田	80	A	
6	友田	65	C+	
7	保田	77	B+	
8	大川	78	B+	
9	木村	32	D	
10	村田	54	D	
11	川上	66	C+	
12				

図 16 得点から成績を求める

- VLOOKUP 関数で照合方法が TRUE(行と行の間も検索する場合) のとき、照合値の列（一番左の列）が昇順（小さい順）でソート（並べ替え）されていなくてはなりません。したがって、表 15 に対応する VLOOKUP の照合範囲の表は、表 15 の行の順番とは逆になります。「0 から 59」の行が 1 行目、「90～」の行が最下行になります。

9 絶対参照と相対参照の混合

図 17 は、ネットショッピングの支払いを各通貨別に集計したものです。換算レートは、2012年3月現在のものです。赤の部分は計算式で求める部分です。この表は、[テキストのページ](#)からダウンロードできます (S104.xlsx(ネットショッピング))。

表には、日付と品名、支払った通貨の種類 (GBP はイギリスポンド, EUR はユーロ, CHF はスイスフラン, USD はアメリカドルをあらわします。), その通貨での金額を記入します。表の右側で、通貨毎に集計し、換算レートをかけて、日本円 (邦貨) の金額を求めます。日付の入力では「8月10日」は「8/10」と入力します。自動的に、「8月10日」に変換されます。

日付としてうまく表示されない場合や、日付から数値の表示形式に変更するときは、右クリックして、セルの書式設定の表示形式のタブで変更します。

	A	B	C	D	E	F	G	H	I
1	ネットショッピング								
2									
3	日付	品名	通貨	金額	GBP	EUR	CHF	USD	
4	8月10日	洋服	GBP	85	85				
5	8月10日	バッグ	GBP	70	70				
6	8月11日	バッグ	EUR	1000		1000			
7	8月13日	洋服	EUR	500		500			
8	8月13日	バッグ	EUR	600		600			
9	8月14日	書籍	EUR	150		150			
10	8月14日	洋服	CHF	150			150		
11	8月14日	書籍	USD	100				100	
12	8月15日	ソフトウェア	USD	130				130	
13		合計			155	2250	150	230	
14		換算レート			132.15	109.88	91.08	83.41	合計
15		邦貨金額(日本円の金額)			20,483	247,230	13,662	19,184	300,560

図 17 ネットショッピング支払い

次に計算式を設定しましょう。まず、E4 について考えます。相対参照で書いてみましょう。

E4 は、もし、この行の通貨が、GBP であつたら、この行の金額の欄の値を表示する。

↓

E4 は、もし、C4 と E3 が等しかったら、D4 を表示し、そうでなかったら、空白を表示する

(空白は、ダブルコーテーション「”」を2つ重ねます)。

↓

E4: =IF(C4=E3,D4,"")

となります。次に、この計算式を複製してみましょう。

複製元: E4 複製先: E4:H12

としてみましょう。間違っていることがわかります。

では、E5,E6 の計算式（表 4 上部）を見てみましょう。E5 の計算式は、IF(C5=E4,D5,"") になっています。しかし、正しい計算式は、IF(C5=E3,D5,"") です。E6,E7 も表のようになります。原因は、E4 の式中の E3 は、下に複製するとき、数字の部分が増えてはいけないのに、番地調整機能によって、増えてしまっているからです。

表 4 絶対参照と相対参照の混合

セル	単純に複製したときの計算式	正しい計算式
E4:	=IF(C4=E3,D4,"")	=IF(C4=E3,D4,"")
E5:	=IF(C5=E4,D5,"")	=IF(C5=E3,D5,"")
E6:	=IF(C6=E5,D6,"")	=IF(C6=E3,D6,"")
E7:	=IF(C7=E6,D7,"")	=IF(C7=E3,D7,"")
E4:	=IF(C4=E3,D4,"")	=IF(C4=E3,D4,"")
F4:	=IF(D4=F3,E4,"")	=IF(C4=F3,D4,"")
G4:	=IF(E4=G3,F4,"")	=IF(C4=G3,D4,"")
H4:	=IF(F4=H3,G4,"")	=IF(C4=H3,D4,"")

E4 の計算式を右に複製したものを表 4 下部に示します。赤の部分が間違っています。これは、C4 の部分と D4 の部分のアルファベットは、横に複製されるときに、増えてはいけないのに番地調整機能により、増えてしまったからです。

複写のとき、番地調整機能を働かせないように「\$」を付けます。上の例では、

E4: =IF(\$C4=E\$3,\$D4,"")

となります。複写しても、Cと3とDは、変化せずいつもCと3とDです。E4を設定したら、他の部分に複写します。

複写元： E4 複写先： E4:H12

合計などを計算しましょう。換算レートは、1 海外通貨あたり日本円でいくらかを示しています。例えば、155 ポンド (GBP) 使い、換算レートが 1 ポンド=132.154 円のときの邦貨金額 (日本円金額) は、155 ポンド × 132.15 円/ポンド = 20,483 円となります。I15 は、邦貨金額の合計です。15 行目はセルの書式設定で、小数点以下を四捨五入しましょう。

通貨記号の欄を変更したり、現在の為替レート (朝日新聞デジタル：為替 - 経済・マネー) に直したりしても正しく動作するか確認しましょう。

10 練習問題

10.1 1 次関数 3 本のグラフ化

1 次関数 $f(x) = ax + b$ で、 a と b を変えた 3 本をグラフ化する表を作成しましょう。図 18 左の B2:D3 に 3 本の 1 次関数の a と b の値を記入しておきます。ただし、計算式を設定する部分は、次のようにします。

- (1) A6 に -10 を入力し、A7 に「=A6+0.5」とします。
- (2) B6 に計算式を設定し、その式を B6:D46 に複写します。B6 の計算式は、絶対参照と相対参照が混合していますので注意してください。
- (3) 図 18 右のような 3 本のグラフ (散布図 (直線)) を作成しましょう。




図 18 $y = ax + b$ のグラフ化のための表とグラフ

(4) それぞれの直線のパラメータである a, b を変化させたとき、どのように変化するかを確かめましょう。

10.2 練習問題 S105-1

図 19 は、ある架空の電話会社の電話料金の計算表です。この表は、[テキストのページ](#)からダウンロードできます (S105.xlsx(電話料金))。

- 10 円で話せる時間は、通話区分、距離で決まります。
- 通話区分は 3 区分：「平日の昼間」(区分 1)、「平日の夜または休日の昼夜」(区分 2)、「平日、休日の深夜」(区分 3)
- 距離は、通話先までの距離で、表の 14 行目の 0 は、0km(隣接) 以上 20km 未満を表しています。また、19 行目の距離 170 の行は、170km 以上すべてを表しています。
- たとえば、通話区分が 2 で、70km の場合、60 の行の通話区分 2 の列 (3 列目) の場合の 10 円で話せる秒数は 45 秒

	A	B	C	D	E	F	G	H	I
1									
2	日	通話区分	距離	通話時間 (秒)	10円で話せる時間	度数 (切り上げ前)	度数 (切り上げ後)	金額	
3	6月10日	1	35	588	45	13.06666667	14	140	
4	6月11日	2	48	627	60	10.45	11	110	
5	6月12日	3	165	37	45	0.822222222	1	10	
6	6月12日	3	165	156	45	3.466666667	4	40	
7	6月13日	2	23	251	60	4.183333333	5	50	
8	6月15日	1	18	1028	90	11.42222222	12	120	
9							合計	470	
10									
11									
12	料金表 (10円で話せる時間)								
13	距離	1 平日昼間	2 夜・休日昼夜	3 深夜早朝					
14	0	90	90	120					
15	20	60	60	75					
16	30	45	60	75					
17	60	26	45	60					
18	100	20	26	45					
19	170	20	26	36					

図 19 電話料金の計算

になります。

- 料金は、まず度数を、

$$\text{度数} = \frac{\text{通話時間 (秒)}}{10 \text{ 円で話せる秒数}}$$

で計算し (F 列), 小数点以下を切り上げます (G 列, 切り上げの計算方法は, 表 3 を参照).

- 1 度数は 10 円なので、度数の列 (G 列) に 10 をかけたものが電話料金になります。

では、図 19 の上方のエリアを使って電話料金の計算表を作成しましょう。

ヒント

- 「10 円で話せる秒数」は、距離と通話区分を使って、料金表 (A14:D19) から VLOOKUP を使って探します。そのとき、通話区分が 1 のとき 2 列目、2 のとき 3 列目 ... であることから、VLOOKUP の列位置は、通話区分のセルの値を使った計算式になります (通話区分の値に 1 加えた数が列番号になっています)。
- 「度数 (切り上げ前)」は、単純に、 $\frac{\text{通話時間 (秒)}}{10 \text{ 円で話せる秒数}}$ を計算したものです。
- 「度数 (切り上げ後)」は、「度数 (切り上げ前)」の値を切り上げたものです。

完成したら、通話区域、通話の種類や時間を変えてみて、正しく計算されているかどうかチェックしましょう。

11 練習問題 S106-1

扶養家族のいない給与所得のみ給与所得者の所得税の計算を行ってみます*1

この表は、[テキストのページ](#)からダウンロードできます (S106.xlsx(所得税の計算))。

11.1 入力エリア

本計算表では、図 20 のように給与収入金額と社会保険料の支払い金額を入力し、最終的に所得税額を求めます。

*1 2020 年度まで日本国の制度に近いものを目指していましたが、2021 年度より、よりシンプルで比較的きれいな旧来部分を残している扶養家族無しの給与所得のみの方の所得税の計算をします。配偶者控除や子ども手当 (児童手当に名称変更) に所得制限が導入されたことにより、所得金額が増えると、所得税などが増えたり、児童手当が削減され、所得制限の影響で手取り金額が減少するということが起きています。

給与収入金額 名目上支払われた給与の総額

社会保険料の支払金額 健康保険，厚生年金，国民年金などの保険料（支払額は，給与収入の 15% 程度です。）

	A	B	C
1	入力エリア		
2	給与収入金額	4,520,000	
3	社会保険料の支払金額	678,000	
4			

図 20 所得税の計算 (入力エリア)

11.2 給与所得控除の計算

給与所得控除は，給与所得者の概算の経費です。所得税は，収入から経費を引いた所得に課税されます。

給与所得控除は，給与収入の多寡でグループ分けをし，そのグループでの率と加算額で，給与所得控除額を計算します。

$$\text{給与所得控除額} = \text{給与収入額} \times \text{率} + \text{加算} \cdot \text{減算額}$$

表 5 は，給与収入額により 6 個のグループに分け，それぞれのグループの率を示しています。

率と加算・減算金額は，VLOOKUP 関数を利用して求めます。図 21 の右の表は，表 5 を VLOOKUP 関数で利用できるように作成した物です。D6,E6,F6 は，給与収入金額が 0 (D6) 以上 1,625,001(D7) 未満の場合，率が 0，加算・減算金額が 550,000 であることを示しています。また D7,E7,F7 は，給与収入金額が 1,625,001 (D7) 以上 1,800,001(D8) 未満の場合，率が 40%，加算・減算額 -100,000(-100,000 を加算)であることを示しています。

給与収入金額 (B6) 入力エリアの給与収入金額の複写 (B2 のセルと同じ値にします。=B2 とします)。

表5 給与所得控除額の率と加算額 (2020年)

給与収入金額	率	加算・減算額
0 ~ 1,625,000	0%	+550,000
1,625,001 ~ 1,800,000	40%	-100,000
1,800,001 ~ 3,600,000	30%	+80,000
3,600,001 ~ 6,600,000	20%	+440,000
6,600,001 ~ 8,500,000	10%	+1,100,000
8,500,001 ~	0%	+1,950,000

	A	B	C	D	E	F	G
4							
5	給与所得控除の計算			給与収入金額	率	加算・減算額	
6	給与収入金額	4,520,000		0	0%	550,000	
7	給与所得控除の率	20%		1,625,001	40%	-100,000	
8	給与所得控除の加算・減算額	440,000		1,800,001	30%	80,000	
9	給与所得控除額	1,344,000		3,600,001	20%	440,000	
10				6,600,001	10%	1,100,000	
11	所得金額(給与収入金額-給与所得控除額)	3,176,000		8,500,001	0%	1,950,000	
12							

図21 給与所得控除の計算

給与所得控除の率 (B7) VLOOKUP 関数を使って求めます。表5に対応する表をあらかじめ作っておきます(図21右)。VLOOKUP関数(本章7節)で、B6の値を使って(照合値)、\$D\$6:\$F\$11(照合範囲)で、2列目の率の列の値を出しま

す。あいだの値も探すので、照合方法は、TRUE です。

照合範囲は、コピーして使うことがあるので絶対参照 (\$付き) にします。

給与所得控除の加算額 (B8) 同様に VLOOKUP 関数を使って求めます (照合範囲は絶対参照)。

給与所得控除額 (B9) 給与所得控除額 = (給与収入金額 × 率) + 加算・減算額 で求めます。

所得金額 (B11) 所得金額 = 給与収入金額 - 給与所得控除額 ですが、給与収入金額が 0 付近のとき、給与所得控除額が 550,000 となり、所得金額が負になります。給与収入金額 - 給与所得控除額 が負の時は、所得金額を 0 にします。

B11 =IF(B6-B9<0,0,B6-B9)

11.3 所得控除の計算

所得控除は、所得税を計算するときに、所得から差し引かれる金額 (課税されない金額) です。ここでは、基礎控除と社会保険料控除のみを扱います。

	A	B	C
13	所得控除の計算		
14	社会保険料控除	678,000	
15	基礎控除	480,000	
16	所得控除額	1,158,000	
17			

図 22 所得控除の計算

社会保険料控除 (B14) 入力エリアの社会保険料の支払額と同じ値 (=を使った式)

基礎控除 (B15) 480,000 (定数として入力)*2

合計 (所得から差し引かれる金額) (B16) 社会保険料控除と基礎控除の合計額

11.4 課税される所得金額の計算

	A	B	C
18	課税される所得金額の計算		
19	課税される所得金額(所得金額 - 所得から差し引かれる金額)	2,018,000	
20	課税される所得金額が負の時は0	2,018,000	
21	課税される所得金額(1000円未満切り捨て後)	2,018,000	
22			

図 23 課税される所得金額の計算

課税される所得金額 (所得金額 - 所得から差し引かれる金額) 所得金額 - 所得から差し引かれる金額 を計算します。

課税される所得金額が負の時は 0 課税される所得金額が負のときは、0 とします。この欄は、IF 関数を使って「課税される所得金額 (所得金額 - 所得から差し引かれる金額)」が負の時、0 としてください。正のときは、「課税される所得金額 (所得金額 - 所得から差し引かれる金額)」と同じ値にしてください。

課税される所得金額 (1000 円未満切り捨て後) 課税される所得金額の 1000 円未満の端数は切り捨てですので、切り捨ての計算 (表 3 参照) をしてください。

*2 所得金額が 2,400 万円を越える場合は、基礎控除額は異なります。

11.5 所得税額の計算

基準所得税額を次式で求めます。

$$\text{基準所得税額} = \text{課税される所得金額} \times \text{税率} - \text{控除額}$$

表 6 所得税額

課税される所得金額	税率	控除額
0 ~ 1,949,000	5%	0
1,950,000 ~ 3,299,000	10%	97,500
3,300,000 ~ 6,949,000	20%	427,500
6,950,000 ~ 8,999,000	23%	636,000
9,000,000 ~ 17,999,000	33%	1,536,000
18,000,000 ~ 39,999,000	40%	2,796,000
40,000,000 ~	45%	4,796,000

税率、控除額は、表 6 のように課税される所得金額により 7 つのグループに分かれ、それぞれ税率、控除額が異なります。税率、所得税控除額は、VLOOKUP 関数を使います。表 6 に対応する Excel の対応表は、図 21 右を参考に作成します。

税率・所得税控除額 表 6 に対応する Excel の対応表を使い、VLOOKUP 関数（照合範囲は絶対参照）を使って求めます。

基準所得税額 課税される所得金額 × 税率 - 控除額 で求めます。

復興特別所得税額 復興特別所得税額は、基準所得税額の 2.1% です。

	A	B	C
23	所得税額の計算		
24	税率(限界税率)	10%	
25	控除額	97,500	
26	基準所得税額	104,300	
27	復興特別所得税額	2,190	
28	所得税額(100円未満切り捨て前)	106,490	
29	所得税額(100円未満切り捨て後)	106,400	
30	手取額(給与収入金額-社会保険料-所得税額)	3,735,600	
31			

図 24 所得税額の計算

所得税額（100 円未満切り捨て前） 基準所得税額と復興特別所得税額の和です。

所得税額（100 円未満切り捨て後） Rounddown 関数を使って求めます。

手取額（給与収入金額 - 社会保険料 - 所得税額） 実際に支払われる金額で手取額とします。ただし、実際には、この他にも地方税の住民税が差し引かれます。

表 7 は、確認用の給与収入金額と社会保険料の支払金額を入力したとき、各項目の値がいくつになるのか示した表です。

11.6 給与収入金額と所得税額・手取額の関係（可視化）

11.5 節までの計算で、給与収入金額と社会保険料の支払額を入力すれば、手取額を計算する表を作成しました。

計算する部分は、VLOOKUP で参照する表を除き、B1~B30 の 1 列で計算するように作成されており、また、VLOOKUP で参照する表は、絶対参照にしています。したがって、B1~B30 の部分を複製して、給与収入金額と社会保険料の支払額を

表7 一部の入力値と最終結果(確認用)

入力エリア	パターン 1	パターン 2	パターン 3	パターン 4
給与収入金額	3,000,000	6,000,000	9,000,000	12,000,000
社会保険料の支払金額	450,000	900,000	1,350,000	1,800,000
計算過程・結果				
給与所得控除の率	30%	20%	0%	0%
給与所得控除の加算額	80,000	440,000	1,950,000	1,950,000
給与所得控除額	980,000	1,640,000	1,950,000	1,950,000
課税される所得金額(1000円未満切り捨て後)	1090,000	2,980,000	5,220,000	7,770,000
税率(限界税率)	5%	10%	20%	23%
所得税額(100円未満切り捨て後)	55,600	204,700	629,400	1,175,200
手取額(給与収入金額-社会保険料-所得税額)	2,494,400	4,895,300	7,020,600	9,024,800

変更すれば、さまざまなパターンを比較できます。

給与収入のある人(労働者)の社会保険料は、ほぼ給与収入額により決まり、その率は15%程度です(社会保険料は、労使折半で、企業などの使用者も15%程度の負担する制度になっています)。そこで、給与収入を0~2000万円程度まで1万円単位で給与収入を変化させ、所得税額と手取額がどのように変化するかを表とそのグラフを作成します。

(1) グラフ用のワークシートの作成 ワークシート「所得税計算」を複製して「収入と手取額グラフ」を作成します。

(2) 所得税を計算する範囲を複製 ワークシート「収入と手取額グラフ」のB1~B30をJ1~J30に複製します。

複製元： B1:B30

複製先： J1:J30

(3) ウィンドウ枠の固定 A 列は J 列以降でも参照するので、A 列を固定します。

メニューの 表示 → ウィンドウ枠の固定 → 先頭列を固定

動画:所得税シミュレーション 1

給与収入額が変化すると、所得税額や手取額がどう変化するかシミュレーションして、グラフ化してみます。社会保険料は給与収入額の 15% と仮定しているので、J2 の給与収入額を指定すれば、J3 の社会保険料はその 15% を計算するようにします。したがって J2 の給与収入額を与えれば、J29 の所得税額と J30 の手取額を 1 列で計算できます。

32 行目から 34 行目に、給与収入額、所得税額、手取額の一覧を作成し、グラフ化などを容易にします。32 行目の給与収入額を 0 から 10 万円ずつ増やして、2000 万円まで、右方向に複写します。

(4) 一覧表の作成 I32,I33,I34 に「給与収入額」、「所得税額」、「手取額」と記入します。

(5) J 列の設定 給与収入額 (J32) を 0 として計算します。J32 の値を J2 に転記し、社会保険料を給与収入額の 0.15 倍にします。J33 に上部で計算した所得税額、J34 に手取額を転記します。

J32:	0	J2:	=J32
J3:	=J2*0.15		
J33:	=J29	J34:	=J30

(6) K 列の設定 給与収入額 を J 列より 100,000 増やします。他は、J 列の計算式と同じとします。

複写元:	J1:J34	複写先:	K1:K34
K32:	=J32+100000		

(6) K 列の計算式を右方向への複写 K 列の計算式を 199 列 (HB 列まで) 複写します。

複製元：

K1:K34

複製先：

L1:HB34

動画:所得税シミュレーション 2

これで、J32:HB34 に給与収入額と所得税額、手取額の変化の表ができました。これをグラフ化して、変化を見ます（図 25）。

動画:所得税シミュレーション 3




図 25 給与収入額と所得税額・手取額 (横軸:給与収入額)

日本の所得税制は所得が増えると、より高い税率を課す累進課税とされています。実際、表 6 を見ると、課税される所

得金額が増えると税率は上昇しています。給与収入に対して、所得税額は累進的に増えるのかを見てみましょう。

下記の動画のように、2,000万円までの所得税額だけのグラフ（図 26）と 500万円までの所得税額だけのグラフ（図 27）を作成してみましょう。




図 26 給与収入額と所得税額 (給与収入額 2000万円まで)

動画:所得税シミュレーション 4

図 26 をみると、給与収入額は増大すると所得税額は増大しています。また、その増大の割合は、給与収入額が増大すると大きくなる、傾きが急になるように見えます。実際、表やグラフより、給与収入額 300 万 → 310 万に増大すると所得税額は 2800 円増大し、給与収入額 400 万 → 410 万に増大すると所得税額は 3300 円増大し、給与収入額 500 万 → 510 万に増大す




図 27 給与収入額と所得税額 (給与収入額 500 万円まで)

ると所得税額は 6600 円増大します。グラフを見ると給与収入が増加すると所得税額のグラフの傾きがだんだん大きくなり、累進課税になっているようで、おおよそ、累進課税であることがわかります。

11.7 給与収入に対して累進性の確認 (参考)

これがすべての給与収入額で満たしているか見てみましょう。図 26 からは、傾きがだんだん急になっているように見えますが、給与収入額が数百万のまでの範囲が見えにくくなっています。そこで、図 27 のように 500 万円までのグラフで比較してみます。

図 27 では、120 万から 160 万の範囲で傾きが急で、160 万から 370 万くらいまでは、緩やかなのよう見えます。表をみ

ると 120 万から 160 万では、給与収入額が 10 万増えると 4300 円程度増えるのに対して、160 万から 360 万くらいまでは、2800 円程度です。狭い範囲ですが、給与収入に対して累進的に課税されない設計になっていることがわかります。

原因は、表 5 の給与所得控除の計算方法が原因で、120 万から 160 万の範囲では、率は 0% となっており、給与収入額が増えても給与所得控除は増えないのに対して、160 万から 360 万くらいまでは、40% または 30% で給与所得控除が増え、所得の増大が抑えられ、所得税額の増大は抑えられています。